
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl"

begin

  ascii_filename = "hmwk1-4-fig.prn"  ; data file for plotting.

  seismic = asciiread(ascii_filename,(/2172,3/),"float")

  x = seismic(:,1)  ; Column 1 of file contains X values.
  y = seismic(:,0)  ; Column 2 of file contains Y values.
  z = seismic(:,2)  ; Column 3 of file contains Z values.

  numxout = 13     ; Define output grid for call to "natgrids".
  numyout = 181
  xmin    = min(x)
  ymin    = min(y)
  xmax    = max(x)
  ymax    = max(y)
  xc      = (xmax-xmin)/(numxout-1)
  yc      = (ymax-ymin)/(numyout-1)
  xo      = xmin + ispan(0,numxout-1,1)*xc
  yo      = ymin + ispan(0,numyout-1,1)*yc

  zo = natgrids(x, y, z, xo, yo)  ; Interpolate.

  xo@long_name = "Month"   ; Define some attributes.
  yo@long_name = "Latitude"
  zo@long_name = "Daily average insolation at TOA"

  xwks   = gsn_open_wks( "x11","fighmwk1") ; Open an X11 workstation.
  pdfwks = gsn_open_wks( "pdf","fighmwk1") ; Open a PDF workstation.

  cmap = (/(/1., 1., 1./), (/0., 0., 0./), (/1., 0., 0./), (/1., 0., .4/), \
           (/1., 0., .8/), (/1., .2, 1./), (/1., .6, 1./), (/.6, .8, 1./), \
           (/.2, .8, 1./), (/.2, .8, .6/), (/.2, .8, 0./), (/.2, .4, .0/), \
           (/.2, .4, .4/), (/.2, .4, .8/), (/.6, .4, .8/), (/.6, .8, .8/), \
           (/.6, .8, .4/), (/1., .6, .8/)/)

  gsn_define_colormap(  xwks,cmap) ; Define a color map for each of the
  gsn_define_colormap(pdfwks,cmap) ; 2 workstations you just opened.

;----------- Begin first plot -----------------------------------------

  resources                       = True
  resources@sfXArray              = xo            ; X axes data points
  resources@sfYArray              = yo            ; Y axes data points
  resources@tmXBMode      = "Explicit"   ; Define your own tick mark labels.
  resources@tmXBLabelFont = 25           ; Change font of labels.
  resources@tmXBLabelFontHeightF = 0.015 ; Change font height of labels.
  resources@tmXBMinorOn   = False        ; No minor tick marks.
  resources@tmXBValues    = ispan(1,12,1); Location to put tick mark labels
                                         ; (13 points with January repeated).
  resources@tmXBLabels    = (/"Jan","Feb","Mar","Apr","May","Jun",\
                              "Jul","Aug","Sep","Oct","Nov","Dec","Jan"/)

  resources@tiMainString          = zo@long_name  ; Main title
  resources@tiMainFont            = "Times-Bold"
  resources@tiXAxisString         = "Month"       ; X axis label.
  resources@tiYAxisString         = "Latitude"    ; Y axis label.

  resources@cnFillOn              = True     ; Turn on contour fill.
  resources@cnInfoLabelOn         = False    ; Turn off info label.
  resources@cnLineLabelsOn        = False    ; Turn off line labels.

  resources@lbOrientation         = "Horizontal" ; Draw it horizontally.
  resources@lbPerimOn             = False        ; Turn off perimeter.
  resources@pmLabelBarDisplayMode = "Always"     ; Turn on a label bar.
  resources@pmLabelBarSide        = "Bottom"     ; Change location of 
                                                 ; label bar.
  resources@vpYF = 0.85   ; Change Y location of plot.

  zo!0 = "i"  ; Name the dimensions of "zo".
  zo!1 = "j"

  contour = gsn_contour(xwks,zo(j|:,i|:),resources) ; Draw a contour plot.

;----------- Begin second plot -----------------------------------------

  delete(resources)   ; Start with a new list of resources.

  resources                     = True
  resources@tiMainString        = ":F26:Slices by Latitude"   ; Define a title.

  resources@xyLineColors        = (/2,8,10,14/)   ; Define line colors.
  resources@xyLineThicknessF    = 3.0             ; Define line thickness.
  resources@tmXBMode      = "Explicit"   ; Define your own tick mark labels.
  resources@tmXBLabelFont = 25           ; Change font of labels.
  resources@tmXBLabelFontHeightF = 0.015 ; Change font height of labels.
  resources@tmXBMinorOn   = False        ; No minor tick marks.
  resources@tmXBValues    = ispan(1,12,1); Location to put tick mark labels
                                         ; (13 points with January repeated).
  resources@tmXBLabels    = (/"Jan","Feb","Mar","Apr","May","Jun",\
                              "Jul","Aug","Sep","Oct","Nov","Dec","Jan"/)

  resources@pmLegendDisplayMode    = "Always"     ; Turn on the drawing
                                                  ; of a legend.
  resources@pmLegendZone           = 0            ; Change the location
  resources@pmLegendOrthogonalPosF = 0.31         ; of the legend
  resources@lgJustification        = "BottomRight"

  resources@pmLegendWidthF         = 0.4          ; Change width and
  resources@pmLegendHeightF        = 0.12         ; height of legend.

  resources@pmLegendSide           = "Top"        ; Change location of 
  resources@lgPerimOn              = False        ; legend and turn off
                                                  ; the perimeter.

  resources@xyExplicitLegendLabels = (/"Lat=-90","Lat=-60","Lat=-30","Lat=0",\
                                        "Lat=30","Lat=60","Lat=90"/)

  resources@vpYF      = 0.90  ; Change size and location of plot.
  resources@vpXF      = 0.18
  resources@vpWidthF  = 0.74
  resources@vpHeightF = 0.74

  resources@trYMaxF = 800  ; Set the maximum value for the Y axes.
  resources@trXMinF = 0    ; Set the maximum value for the Y axes.

  xy1 = gsn_xy(xwks,xo,zo(j|0:180:30,i|:),resources) ; Draw an XY plot.

;----------- Begin third plot -----------------------------------------

  delete(resources)   ; Start with a new list of resources.

  resources                     = True
  resources@tiMainString        = ":F26:Month = June"   ; Define a title.

  resources@xyLineColors        = (/2,8,10,14/)   ; Define line colors.
  resources@xyLineThicknessF    = 3.0             ; Define line thickness.

  resources@pmLegendZone           = 0            ; Change the location
  resources@pmLegendOrthogonalPosF = 0.31         ; of the legend
  resources@lgJustification        = "BottomRight"

  resources@pmLegendWidthF         = 0.4          ; Change width and
  resources@pmLegendHeightF        = 0.12         ; height of legend.

  resources@pmLegendSide           = "Top"        ; Change location of 
  resources@lgPerimOn              = False        ; legend and turn off
                                                  ; the perimeter.

  resources@vpYF      = 0.90  ; Change size and location of plot.
  resources@vpXF      = 0.18
  resources@vpWidthF  = 0.74
  resources@vpHeightF = 0.74

  xy2 = gsn_xy(xwks,yo,zo(j|:,i|5),resources) ; Draw an XY plot.

;----------- Begin 4th plot -----------------------------------------

  delete(resources)   ; Start with a new list of resources.

  resources                     = True
  resources@tiMainString        = ":F26:Month = September"   ; Define a title.

  resources@xyLineColors        = (/2,8,10,14/)   ; Define line colors.
  resources@xyLineThicknessF    = 3.0             ; Define line thickness.
  
  resources@pmLegendZone           = 0            ; Change the location
  resources@pmLegendOrthogonalPosF = 0.31         ; of the legend
  resources@lgJustification        = "BottomRight"

  resources@pmLegendWidthF         = 0.4          ; Change width and
  resources@pmLegendHeightF        = 0.12         ; height of legend.

  resources@pmLegendSide           = "Top"        ; Change location of 
  resources@lgPerimOn              = False        ; legend and turn off
                                                  ; the perimeter.
 
  resources@trXMinF = -90    ; Set the maximum value for the Y axes.

  xy3 = gsn_xy(xwks,yo,zo(j|:,i|8),resources) ; Draw an XY plot.

;----------- Draw to other workstations  ------------------------------

  NhlChangeWorkstation(xy1,pdfwks)       ; Change the workstation that the
  NhlChangeWorkstation(contour,pdfwks)   ; contour and XY plot is drawn to.
  NhlChangeWorkstation(xy2,pdfwks)
  NhlChangeWorkstation(xy3,pdfwks)
  draw(contour)                          ; Draw the contour plot to the new
  frame(pdfwks)                          ; workstation and advance the frame.
  draw(xy1)
  frame(pdfwks)
  draw(xy2)
  frame(pdfwks)
  draw(xy3)
  frame(pdfwks)

  delete(xy3)
  delete(xy2)       ; Clean up.
  delete(xy1)
  delete(contour)
end




